
Introduction to the FACTEX Procedure

Overview

The FACTEX procedure constructs orthogonal factorial experimental designs. These designs can be either
full or fractional factorial designs, and they can be with or without blocks. You can also construct designs for
experiments with multiple stages, such as split-plot (Huang, Chen, and Voelkel 1998) and split-lot designs
(Butler 2004). After you have constructed a design by using the FACTEX procedure and run the experiment,
you can analyze the results with a variety of SAS procedures including the GLM and REG procedures.

Factorial experiments are useful for studying the effects of various factors on a response. Texts that discuss
experimental design include Box, Hunter, and Hunter (1978), Cochran and Cox (1957), Montgomery (1991),
and Wu and Hamada (2000). For details about the general mathematical theory of orthogonal factorial
designs, refer to Bose (1947).

NOTE : For two-level designs, instead of using PROC FACTEX directly, a more appropriate tool might
be the ADX Interface for Design of Experiments. The ADX Interface is designed primarily for engineers
and researchers who require a point-and-click solution for the entire experimental process, from building
the designs through determining signi�cant effects to optimization and reporting. ADX gives you most
of the two-level designs provided by the FACTEX procedure in a system that integrates construction and
analysis of designs, without the need for programming. In addition to two-level designs for standard models
(with and without blocking), ADX makes it easy to use PROC FACTEX to construct designs for estimating
particular effects of interest. Moreover, ADX also uses the OPTEX procedure to construct two-level designs
of nonstandard sizes. For more information, seeGetting Started with the SAS ADX Interface for Design of
Experiments.

Features

There is no inherent limit to the number of factors and the size of the design that you can construct with
the FACTEX procedure. Instead of looking up designs in an internal table, the FACTEX procedure uses a
general algorithm to search for the construction rules for a speci�ed design.

You can use the FACTEX procedure to generate designs such as the following:

• factorial designs, such as23 designs, with and without blocking

• fractional factorial designs, such as24� 1
IV , with and without blocking

• split-plot and fractional split-plot designs

• three-level designs, with and without blocking

• mixed-level factorial designs, such as4 � 3 designs, with and without blocking

• randomized complete block design

• factorial designs with outer arrays

• hyper-Graeco-Latin square designs

You can also create more complex designs, such as incomplete block designs, by using the FACTEX procedure
in conjunction with the DATA step.

You can save the design constructed by the FACTEX procedure in a SAS data set. After you have run your
experiment, you can add the values of the response variable and use the GLM procedure to perform analysis
of variance and study signi�cance of effects.

The FACTEX procedure is an interactive procedure. After specifying an initial design, you can submit
additional statements without reinvoking the procedure. After you have constructed a design, you can do the
following:

• print the design points

• examine the alias structure for the design

• modify the design by changing its size, changing the use of blocking, or specifying the effects of
interest in the model again

• output the design to a data set

• examine the confounding rules that generate the design

• randomize the design

• replicate the design

• recode the design from standard values (such as� 1) to values appropriate for your situation

• �nd another design

Learning about the FACTEX Procedure

To learn the basic syntax of the FACTEX procedure, read the section “Getting Started” on page 610, which
contains some simple introductory examples. The summary tables in the section “Summary of Functions” on
page 617 provide an overview of the syntax. The section “Summary of Designs” on page 619 shows simple
ways to construct full factorial designs and fractional factorial designs. The section “Advanced Examples”
on page 634 illustrates construction of complex designs.

Getting Started

The following introductory examples illustrate the capabilities of the FACTEX procedure. See the section
“Advanced Examples” on page 634 for illustrations of complex features.

Example of a Two-Level Full Factorial Design

NOTE : SeeTwo-Level Full Factorial Designin the SAS/QC Sample Library.

This example introduces the basic syntax used with the FACTEX procedure.

An experimenter is interested in studying the effects of three factors—cutting speed (Speed), feed rate
(FeedRate), and tool angle (Angle)—on the surface �nish of a metallic part and decides to run a complete
factorial experiment with two levels for each factor as follows:

Factor Low Level High Level

Cutting speed 300 500
Feed rate 20 30
Tool angle 6 8

This is a23 factorial design—in other words, a complete factorial experiment with three factors, each at two
levels. Hence there are eight runs in the experiment. Since complete factorial designs have full resolution, all
of the main effects and interaction terms can be estimated. For a de�nition of the design resolution, see the
section “Resolution” on page 678.

You can use the following statements to create the required design:

proc factex;
factors Speed FeedRate Angle;
examine design;

run;

These statements invoke the FACTEX procedure, list factor names, and display the generated design points.
By default, the FACTEX procedure assumes that the size of the design is a full factorial and that each factor
has only two levels.

After you submit the preceding statements, you see the following messages in the SAS log:

NOTE: No design size specified.
Default is a full replicate in 8 runs.

NOTE: Design has 8 runs, full resolution.

The output is shown in Figure 7.1. The two factor levels are represented by the coded values –1 and +1.

Figure 7.1 23 Factorial Design

If you prefer to work with the actual (decoded) values of the factors, you can specify these values in an
OUTPUT OUT= statement, as follows:

proc factex;
factors Speed FeedRate Angle;
output out=SavedDesign

Speed nvals=(300 500)
FeedRate nvals=(20 30)
Angle nvals=(6 8);

run;
proc print;
run;

The OUTPUT statement in PROC FACTEX recodes the factor levels and saves the constructed design in the
SavedDesign data set. Since the levels in this example are of numeric type, you use the NVALS= option to
list the factor levels. Optionally, you can use the CVALS= option for levels of character type (see the section
“Example of a Full Factorial Design in Two Blocks” on page 612). The design is saved in a user-speci�ed
output data set (SavedDesign). This is veri�ed by the following message in the SAS log:

NOTE: The data set WORK.SAVEDDESIGN has 8 observations
and 3 variables.

Figure 7.2 shows a listing of the data setSavedDesign.

Figure 7.2 23 Factorial Design after Decoding

Although small complete factorial designs are not dif�cult to create manually, you can easily extend this
example to construct a design with many factors.

Example of a Full Factorial Design in Two Blocks

NOTE : SeeFull Factorial Design in Two Blocksin the SAS/QC Sample Library.

The previous example illustrates a complete factorial experiment that involves eight runs and three factors:
cutting speed (Speed), feed rate (FeedRate), and tool angle (Angle).

Now, suppose two machines (A and B) are used to complete the experiment, with four runs being performed
on each machine. As there is the possibility that the machine affects the part �nish, you should consider
machine as a block factor and account for the block effect in assigning the runs to machines.

The following statements construct a blocked design:

proc factex;
factors Speed FeedRate Angle;
blocks nblocks=2;
model resolution=max;
examine design;

run;

The FACTORS statement in PROC FACTEX speci�es three factors of a23 factorial. The BLOCKS statement
speci�es that the number of blocks is 2. The RESOLUTION=MAX option in the MODEL statement speci�es
a design with the highest resolution—that is, the best design in a general sense. Optionally, if you know
the resolution of the design, you can replace RESOLUTION=MAX with RESOLUTION=r, wherer is the
resolution number. For information about resolution, see the section “Resolution” on page 678.

By default, the FACTEX procedure assumes that the size of the design is a full factorial and that each factor
is at two levels.

After you submit the preceding statements, you see the following messages in the SAS log:

NOTE: No design size specified.
Default is a full replicate in 8 runs.

NOTE: Design has 8 runs in 2 blocks of size 4,
resolution = 6.

The output is shown in Figure 7.3. Note that, by default, the name for the block variable is BLOCK and its
levels are 1 and 2. Also, note that the default factor levels for a two-level design are –1 and 1.

Figure 7.3 23 Factorial Design in Two Blocks before Decoding

You can rename the block variable and use actual levels for the block variable appropriate for your situation
as follows:

proc factex;
factors Speed FeedRate Angle;
blocks nblocks=2;
model resolution=max;
output out=BlockDesign

Speed nvals=(300 500)
FeedRate nvals=(20 30)
Angle nvals=(6 8)
blockname=Machine cvals=(�A� �B�);

run;

proc print;
run;

Figure 7.4 shows the listing of the design saved in the data setBlockDesign.

Figure 7.4 23 Factorial Design in Two Blocks after Decoding

Example of a Half-Fraction Factorial Design

NOTE : SeeHalf-Fraction Factorial Designin the SAS/QC Sample Library.

Often you do not have the resources for a full factorial design. In this case, a fractional factorial design is a
reasonable alternative, provided that the effects of interest can be estimated.

Box, Hunter, and Hunter (1978) describe a fractional factorial design for studying a chemical reaction to
determine what percentage of the chemicals responded in a reactor. The researchers identi�ed the following
�ve treatment factors that were thought to in�uence the percentage of reactant:

• the feed rate of the chemicals (FeedRate), ranging from 10 to 15 liters per minute

• the percentage of the catalyst (Catalyst), ranging from 1% to 2%

• the agitation rate of the reactor (AgitRate), ranging from 100 to 120 revolutions per minute

• the temperature (Temperature), ranging from 140 to 180 degrees Celsius

• the concentration (Concentration), ranging from 3% to 6%

The complete25 factorial design requires 32 runs, but the experimenter can afford only 16 runs.

Suppose that all main effects and two-factor interactions are to be estimated. An appropriate design for this
situation is a design of resolution 5 (denoted as25� 1

V ), in which no main effect or two-factor interaction
is aliased with any other main effect or two-factor interaction but in which two-factor interactions are
aliased with three-factor interactions. This design loses the ability to estimate interactions between three or
more factors, but this is usually not a serious loss. For more information about resolution, see the section
“Resolution” on page 678.

You can use the following statements to construct a 16-run factorial design that has �ve factors and resolution
5:

proc factex;
factors FeedRate Catalyst AgitRate Temperature Concentration;
size design=16;
model resolution=5;
output out=Reaction FeedRate nvals=(10 15)

Catalyst nvals=(1 2)
AgitRate nvals=(100 120)
Temperature nvals=(140 180)
Concentration nvals=(3 6);

run;
proc print;
run;

The design saved in theReaction data set is listed in Figure 7.5.

Figure 7.5 Half-Fraction of a 25 Design for Reactors

The use of a half-fraction causes some interaction terms to be confounded with each other. You can use
the EXAMINE statement with the ALIASING option to determine which interaction terms are aliased, as
follows:

proc factex;
factors FeedRate Catalyst AgitRate Temperature Concentration;
size design=16;
model resolution=5;
examine aliasing;

run;

The alias structure summarizes the estimability of all main effects and two- and three-factor interactions.
Figure 7.6 indicates that each of the three-factor interactions is confounded with a two-factor interaction.
Thus, if a particular three-factor interaction is believed to be signi�cant, the aliased two-factor interaction
cannot be estimated with this half-fraction design.

Figure 7.6 Alias Structure of Reactor Design

When you submit the preceding statements, the following message is displayed in the SAS log:

NOTE: Design has 16 runs, resolution = 5.

This message con�rms that the design exists. If you specify a factorial design that doesnot exist, an error
message is displayed in the SAS log. For instance, suppose that you replaced the MODEL statement in the
preceding example with the following statement:

model resolution=6;

Since the maximum resolution of a25� 1 design is 5, the following message appears in the SAS log:

ERROR: No such design exists.

In general, it is good practice to check the SAS log to see if a design exists.

Using the FACTEX Procedure Interactively

By using the FACTEX procedure interactively, you can quickly explore many design possibilities. The
following steps provide one strategy for interactive use:

1 Invoke the procedure by using the PROC FACTEX statement, and use a FACTORS statement to identify
factors in the design.

2 For a design that involves blocking, use the BLOCKS and MODEL statements. You might want to use the
optimization features for the BLOCKS statement.

3 For a fractional replicate of a design, use the SIZE and MODEL statements to specify the characteristics of
the design. If the design involves blocking, use a BLOCKS statement as well. If you are unsure of the size
of the design or of the number of blocks, use the optimization features for either the BLOCKS or SIZE
statement.

4 Enter a RUN statement and check the SAS log to see if the design exists. If a design exists, go on to the
next step; otherwise, modify the characteristics given in the SIZE, BLOCKS, and MODEL statements.

5 Examine the alias structure of the design. If it is not appropriate for your situation, go back to step 2 and
search for another design.

6 After you have repeated steps 2, 3, and 4 and found an acceptable design, use the OUTPUT statement to
save the design. You can optionally recode factor values, recode and rename the block factor, and create
new factors by using output-value settings.

Details of the FACTEX Procedure

Syntax

You can specify the following statements with the FACTEX procedure. Items within the brackets<> are
optional.

PROC FACTEX < options > ;
FACTORS factor-names < / option > ;
SIZE size-speci�cation ;
MODEL model-speci�cation < / < MINABS < (d) > > < MAXCLEAR < (d) > > > ;
BLOCKS block-speci�cation ;
UNITEFFECT uniteffect / < WHOLE=() > < SUB=() > ;
EXAMINE < options > ;
OUTPUT OUT=SAS-data-set < options > ;

To generate a design and save it in a data set, you use at least the PROC FACTEX, FACTORS, and OUTPUT
statements. The FACTORS statement should immediately follow the PROC FACTEX statement. You use the
MODEL and SIZE statements for designs that are less than a full replicate (for example, fractional factorial
designs). You can use the BLOCKS statement for designs that involve blocking. The EXAMINE statement
can be used as needed.

Summary of Functions

Table 7.1 to Table 7.4 classify the statements and options in PROC FACTEX by function.

Table 7.1 Summary of Options for Specifying the Design

Function Statement Option

Factor Speci�cation
Factor names FACTORSfactor 1 : : : factor f
Number of levels FACTORS factor 1 : : : factor f / NLEV=q
Design Size Speci�cation
(one of the following)
Number of runs SIZE DESIGN=n

Table 7.1 continued

Function Statement Option

Fraction of one full replicate SIZE FRACTION=h
Number ofrun indexing factors SIZE NRUNFACS=m
Minimum number of runs SIZE DESIGN=MINIMUM

or FRACTION=MAXIMUM
or NRUNFACS=MINIMUM

Block Speci�cation
(one of the following)
Number of blocks BLOCKS NBLOCKS=b
Block size BLOCKS SIZE=k
Number ofblock pseudo-factors BLOCKS NBLKFACS=s
Minimum block size BLOCKS NBLOCKS=MAXIMUM

or SIZE=MINIMUM
or NBLKFACS=MAXIMUM

Model Speci�cation
(one of the following)
Estimated effects MODEL ESTIMATE=(effects)
Estimated effects and MODEL ESTIMATE=(effects)

nonnegligible effects NONNEG=(nonnegligible-effects)
Design resolution number MODEL RESOLUTION=r
Design with highest resolution MODEL RESOLUTION=MAXIMUM
Minimum aberration design MODEL EST=(. . . ) <NONNEG=(. . . )> or RES=. . .
(up todth-order interactions) / MINABS<(d)>

Table 7.2 Summary of Options for Searching the Design

Function Statement Option

Search for the Design
Allow maximum time oft seconds PROC FACTEX SECONDS=t or TIME=t
Limit the design searches PROC FACTEX NOCHECK

Table 7.3 Summary of Options for Replicating and Randomizing
the Design

Function Statement Option

Replication
Replicate entire designc times OUTPUT OUT=SAS-data-set DESIGNREP=c
Replicate design for each point OUTPUT OUT=SAS-data-set DESIGNREP=SAS-data-set

in the data set
Replicate each point in design OUTPUT OUT=SAS-data-set POINTREP=p

p times
Replicate data set for each point OUTPUT OUT=SAS-data-set POINTREP=SAS-data-set

in the design



Syntax F 619

Table 7.3 continued

Function Statement Option

Randomization
Randomize the design OUTPUT OUT=SAS-data-set RANDOMIZE
Randomize the design but not the OUTPUT OUT=SAS-data-set RANDOMIZE NOVALRAN

assignment of factor levels
Specify seed number OUTPUT OUT=SAS-data-set RANDOMIZE .u/

Table 7.4 Summary of Options for Examining and Saving the
Design

Function Statement Option

List the Design
Coded factor and block levels EXAMINE DESIGN
List the Design Characteristics
Alias structure EXAMINE ALIASING<(d)>
(up todth-order interactions)
Confounding rules EXAMINE CONFOUNDING
Save the Design
Coded factor levels OUTPUT OUT=SAS-data-set
Decoded factor levels OUTPUT OUT=SAS-data-set factor-name
(numeric type) NVALS=(level1: : : levelq)
Decoded factor levels OUTPUT OUT=SAS-data-set factor-name
(character type) CVALS=('level1' : : : 'levelq')
Block variable name OUTPUT OUT=SAS-data-set BLOCKNAME=block-name
Decoded block levels OUTPUT OUT=SAS-data-set BLOCKNAME=block-name
(numeric type) NVALS=(level1: : : levelb)
Decoded block levels OUTPUT OUT=SAS-data-set BLOCKNAME=block-name
(character type) CVALS=('level1' : : : 'levelb')

Summary of Designs

Table 7.5 summarizes basic design types that you can construct with the FACTEX procedure by providing
example code for each type.

Table 7.5 Basic Designs Constructed by the FACTEX Procedure

Design Type Example Statements

A full factorial design in three fac-
tors, each at two levels coded as –1
and +1.

proc factex;
factors Pressure Temperature Time;
examine design;

run;

Table 7.5 continued

Design Type Example Statements

A full factorial design in three fac-
tors, each at three levels coded as –1,
0, and +1.

proc factex;
factors Pressure Temperature Time / nlev= 3;
examine design;

run;

A full factorial design in three fac-
tors, each at two levels. The entire
design is replicated twice, and the
design with recoded factor levels is
saved in a SAS data set.

proc factex;
factors Pressure Temperature Time;
output out= SavedDesign designrep= 2

Pressure cvals=(�low� �high�)
Temperature nvals=(200 300)
Time nvals=(10 20);

run;

A full factorial design in three fac-
tors, each at two levels coded as –1
and +1. Each run in the design is
replicated three times, and the repli-
cated design is randomized and saved
in a SAS data set.

proc factex;
factors Pressure Temperature Time;
output out= SavedDesign

pointrep= 3 randomize;
run;

A full factorial design in three con-
trol factors, each at two levels coded
as –1 and +1. A noise factor design
(outer array) read from a SAS data
set is replicated for each run in the
control factor design (inner array),
and the product design is saved in a
SAS data set.

proc factex;
factors+ Pressure Temperature Time;

output out =+ SavedDesign
pointrep=+ OutArray;

run;

A full factorial blocked design in
three factors, each at two levels
coded as –1 and +1. The design is
arranged in two blocks and saved in
a SAS data set. By default, the block
variable is named BLOCK and the
two block levels are numbered 1 and
2.

proc factex;
factors Pressure Temperature Time;
blocks nblocks= 2;
output out= SavedDesign;

run;



Statement Descriptions F 621

Table 7.5 continued

Design Type Example Statements

A full factorial blocked design in
three factors, each at two levels
coded as –1 and +1. Each block con-
tains four runs; the block variable is
renamed and the block levels of char-
acter type are recoded. The design is
saved in a SAS data set.

proc factex;
factors Pressure Temperature Time;
blocks size= 4;
output out= SavedDesign

blockname= Machine cvals=(�A� �B� );
run;

A fractional factorial design of reso-
lution 4 in four factors, each at two
levels coded as –1 and +1. The size
of the design is eight runs.

proc factex;
factors Pressure Temperature Time Catalyst;
size design= 8;
model resolution= 4;
examine design;

run;

A one-half fraction of a factorial de-
sign in four factors, each at two lev-
els coded as –1 and +1. The design
is of maximum resolution. The de-
sign points, the alias structure, and
the confounding rules are listed.

proc factex;
factors Pressure Temperature Time Catalyst;
size fraction= 2;
model resolution=maximum;
examine design aliasing confounding;

run;

A one-quarter fraction of a factorial
design in six factors, each at two lev-
els coded as –1 and +1. Main effects
are estimated, and some two-factor
interactions are considered nonnegli-
gible. The design is saved in a SAS
data set.

proc factex;
factors x1-x6;
size fraction= 4;
model estimate=( x1 x2 x3 x4 x5 x6 )

nonneg =( x1 * x5 x1 * x6 x5 * x6 );
output out = SavedDesign;

run;

Statement Descriptions

This section provides detailed syntax information for the FACTEX procedure statements, beginning with the
PROC FACTEX statement. The remaining statements are presented in alphabetical order.

PROC FACTEX Statement

PROC FACTEX < options > ;

You use the PROC FACTEX statement to invoke the FACTEX procedure. The followingoptionsare available:

NAMELEN
speci�es the length of effect names in tables and output data sets to ben characters long, wheren is a
value between 20 and 200 characters. The default length is 20 characters.

NOCHECK
suppresses a technique for limiting the amount of search required to �nd a design. The technique
dramatically reduces the search time by pruning branches of the search tree that are unlikely to contain
the speci�ed design, but in rare cases it can keep the FACTEX procedure from �nding a design that
does in fact exist. The NOCHECK option turns off this technique at the potential cost of an increase
in run time. Note, however, that the run time is always bounded by the TIME= option or its default
value. For more information about the NOCHECK option, see the section “Speeding Up the Search”
on page 691.

TIME=t

SECONDS=t
speci�es the maximum number of seconds to spend on the search. The default is 60 seconds.

BLOCKS Statement

BLOCKS block-speci�cation ;

You use the BLOCKS statement to specify the blocks or split-plot units in the design. By default, the
FACTEX procedure constructs designs that do not contain blocks. If you use the BLOCKS statement, you
also need to use the MODEL statement or SIZE statement. In particular, if you use the BLOCKS statement
and your design is a fractional factorial design, you must use the MODEL statement.

The two simplest explicitblock-speci�cationsthat you can use are as follows:

• NBLOCKS=b, which speci�es the number of blocks (b) in the design

• SIZE=k, which speci�es the number of runs (k) in each block

Use only one of these two options. In all, there are seven mutually exclusiveblock-speci�cationsthat you
can use, as described by the following list.

NBLKFACS= s
speci�es the number of block pseudo-factors for the design. The design contains a different block for
each possible combination of the levels of the block pseudo-factors. Values ofs are the integers 1, 2,
and so on. See the section “Block Size Restrictions” on page 624 for details.

If each factor in the design hasq levels, then NBLKFACS=s speci�es a design withqs blocks. The
size of each block depends on the number of runs in the design, as speci�ed in the SIZE statement. If
the design hasn runs, then each block hasn=qs runs.

The following statement illustrates how to request a two-level factorial design arranged in eight (23)
blocks:

blocks nblkfacs=3;

For more information about pseudo-factors, see the section “Types of Factors” on page 676.

NBLOCKS= b
speci�es the number of blocks in the design. The values ofb must be a power ofq, the number of
levels of each factor in the design. See the section “Block Size Restrictions” on page 624 for details.
The size of each block depends on the number of runs in the design, as speci�ed in the SIZE statement.
If the design hasn runs, then each block hasn=b runs. See the section “Example of a Full Factorial
Design in Two Blocks” on page 612 for an illustration of this option.

The following statement illustrates how to specify a design arranged in four blocks:

blocks nblocks=4;

SIZE=k
speci�es the number of runs per block in the design. The valuek must be a power ofq, the number of
levels for each factor in the design. The number of blocks depends on the number of runs in the design,
as speci�ed in the SIZE statement. If the design hasn runs, then it hasn=k blocks.

CAUTION: Do not confuse the SIZE= option in the BLOCKS statement with the SIZE statement,
which you use to specify the overall size of the design. See the section “SIZE Statement” on page 633
for details of the SIZE statement.

The following statement illustrates how to specify blocks of size two:

blocks size=2;

NBLKFACS=MAXIMUM

NBLOCKS=MAXIMUM

SIZE=MINIMUM
constructs a blocked design with the minimum number of runs per block, given all the other character-
istics of the design. In other words, the block size is optimized. You cannot specify this option if you
specify any of the design size optimization options in the SIZE statement (see DESIGN=MINIMUM).

UNITS=(units-speci�cations )
speci�es unit factors that index the runs of the experiment. Eachunit-speci�cationhas the form

unitfactor = number-of-levels

where the number of levels for each unit factor must be a power of the number of levels speci�ed in the
FACTORS statement (2 by default). You can give multipleunit-speci�cationsin the UNITS= option;
the product of their numbers of levels must be less than the size of the experiment, as speci�ed in the
SIZE statement.

Unit factors are not involved in the model structure of the design. Instead, you use a UNITS=() blocks
speci�cation in conjunction with one or more UNITEFFECT statements to constrain how the factor
levels can change across the runs of the experiment.

The following statement illustrates how to specify two unit factors:

blocks units=(Unit1=4 Unit2=8);

See the section “Split-Plot Designs” on page 686 for details about how to use the UNITS= option and
the UNITEFFECT statement to construct split-plot designs.

Equivalent BLOCK Speci�cations
The three explicitblock-speci�cationsare related to each other, as demonstrated by the following example.

Suppose you want to construct a design for 11 two-level factors in 128 runs in blocks of size 8. Since
128=24 D 128=16D 8, three equivalent block speci�cations are as follows:

blocks nblkfacs=4;
blocks nblocks=16;
blocks size=8;

Block Size Restrictions
The number of blocks and the number of runs in each block must be less than the total number of runs in the
design. Hence, there are some restrictions on the block size.

• If you use SIZE=k or NBLOCKS=b, the numbers you specify fork andb must be less than or equal to
the size of the design, as speci�ed in the SIZE statement. Or, if you do not use a SIZE statement,k and
b must be less than or equal to the number of runs for a full replication of all possible combinations of
the factors.

For example, for a23 design you cannot specify a design arranged in 8 blocks (NBLOCKS=8).
Likewise, you cannot construct a design with block size greater than 8 (SIZE=8).

• If you use NBLKFACS=s, the value ofscan be no greater than the number of run-indexing factors,
which give the number of runs needed to index the design. For details, see “Types of Factors” on
page 676 and “Theory of Orthogonal Designs” on page 687.

UNITEFFECT Statement

UNITEFFECT uniteffect / < WHOLE=(wholeuniteffects) > < SUB=(subuniteffects) > ;

You use the UNITEFFECT statement to specify constraints on how the factor levels can change across the
runs of the experiment. Such constraints are known as randomization restrictions. UNITEFFECT statements
are used in conjunction with a UNITS speci�cation in the BLOCKS statement, where unit factors are de�ned
that index the runs of the experiment.

Theuniteffectis an interaction betweenunitfactorsde�ned in the BLOCKS UNITS=() speci�cation:

unitfactor * . . . * unitfactor

It de�nes a partition of the runs on which to apply whole-unit and subunit effects of the factors named in the
FACTORS statement.

Thewholeuniteffectsspeci�ed by the WHOLE=() option typically de�ne a necessary feature of how the
experiment must be designed, and are thus known as “design constraints.” In contrast, the SUB=() option
indicates which unit mean contrasts will be used to compute thesubuniteffectsand which random error
terms will be used to test them, and are thus known as “model constraints.” For bothwholeuniteffectsand
subuniteffects, the effects listed must be enclosed within parentheses, as in the MODEL statement. See the
section “Specifying Effects in the MODEL Statement” on page 677 for details.

If you have speci�ed units in the BLOCK statement as

blocks units=(WholePlot=4);

then the following statement illustrates how to specify unit effects corresponding to these units:

uniteffect WholePlot / whole=(x1-x3) sub=(x4-x6);

See the section “Split-Plot Designs” on page 686 for details about how to use the UNITS= option and the
UNITEFFECT statement to construct split-plot designs.

EXAMINE Statement

EXAMINE < options > ;

You use the EXAMINE statement to specify the characteristics of the design that are to be listed in the output.

Theoptionsare remembered by the procedure; once speci�ed, they remain in effect until you submit a new
EXAMINE statement with different options or until you turn off all EXAMINE options by submitting just

examine;

The followingoptionsare available.

ABERRATION

AB
displays the aberration vector for the design, summarizing the confounded interactions. See the section
“Minimum Aberration” on page 684 for more information.

ALIASING < ( < d > < UNITS < = ONE | ALL > > ) >

A < ( < d > < UNITS < = ONE | ALL > > ) >
lists the alias structure of the design, which identi�es effects that are confounded with one another and
are thus, indistinguishable.

You can specifyd in parentheses immediately after the ALIASING option for a listing of the alias
structure with effects up to and including orderd . For example, the following statement requests
aliases for up to fourth-order effects (for example, A*B*C*D):

examine aliasing(4);

Each line of the alias structure is listed in the form

effect = effect = . . . = effect

for as many effects as are aliased with one another.

The default value ford is determined automatically from the model as follows:

• If you specify the model with a resolution numberr in the MODEL statement, thend is the
integer part of.r C 1/=2.

• If you specify the model with a list of effects in the MODEL statement, thend is the larger of the
following, where main effects have order 1, two-factor interactions have order 2, and so on:

– one plus the largest order of an effect to be estimated

– the largest order of an effect considered to be nonnegligible

If your design involves unit effects, then you can also specify UNITS in parentheses immediately after
the ALIASING option to display for each treatment effect alias string the top unit effect with which it
is aliased. Using the UNITS option can give you information about which error stratum can be used to

estimate the background error variance for each estimable treatment effect. If unit effects are nested, as
they typically are in complex split-plot designs, then treatment effects can be aliased with more than
one unit effect. In this case, specify (UNITS=ALL) to see all unit effects with which each treatment
effect is aliased.

For details about aliasing, see the section “Alias Structure” on page 684.

CONFOUNDING

C
lists the confounding rules used to construct the design. For the de�nition of confounding rules, see
the sections “Confounding Rules” on page 683 and “Suitable Confounding Rules” on page 688.

DESIGN

D
lists the points in the design in standard order with the factor levels coded. For a description of the
randomization and coding rules, see the section “OUTPUT Statement” on page 629.

SUMMARY < (< d >) >

S < (< d >) >
displays the modeling summary for the design, summarizing how many interactions of each order
are estimable and how many are clearly estimable (that is, unaliased with any other interactions of
interest).

You can specifyd in parentheses immediately after the SUMMARY option to display the modeling
summary accounting for effects up to and including orderd . The default value ford is determined
automatically from the model as it is for the ALIASING option.

FACTORS Statement

FACTORS factor-names < / option > ;

You use the FACTORS statement to start the construction of a new design by naming the factors in the design.
The FACTORS statement clears all previous speci�cations for the design (number of runs, block size, and so
on). Use it when you want to start a new design.

NOTE : The FACTORS statement should be the �rst statement following the PROC FACTEX statement.

In the FACTORS statement,

factor-names
lists names for the factors in the design. These names must be valid SAS variable names. See the
section “Types of Factors” on page 676 for details.

The followingoptionis available:

NLEV=q
speci�es the number of levels for each factor in the design. The value ofq must be an integer greater
than or equal to 2. The default value forq is 2. In order to construct a design that involves either
fractionation or blocking,q must be either a prime number or an integer power of a prime number. For
the reason behind this restriction, see the section “Structure of General Factorial Designs” on page 687.

MODEL Statement

MODEL model-speci�cation < / < MINABS < (d) > > < MAXCLEAR < (d) > > > ;

You use the MODEL statement to provide the model for the construction of the factorial design. The model
can be speci�ed either directly by specifying the effects to be estimated with the ESTIMATE= option or
indirectly by specifying the resolution of the design with the RESOLUTION= option.

NOTE : If you create a fractional factorial design or if you create a design that involves blocking, the MODEL
statement is required.

The twomodel-speci�cationsare described as follows:

ESTIMATE=(effects) < option >
identi�es theeffectsthat you want to estimate with the design. To specifyeffects, simply list the names
of main effects, and join terms in interactions with asterisks. Theeffectslisted must be enclosed within
parentheses. See the section “Specifying Effects in the MODEL Statement” on page 677 for details.
You can use EST or E for the keyword ESTIMATE.

After the ESTIMATE=option, you can specify the followingoption:

NONNEGLIGIBLE=( nonnegligible-effects)
identi�es nonnegligible effects. These are the effects whose magnitudes are unknown, but you do not
necessarily want to estimate them with the design. If you do not want certain effects to be aliased with
ESTIMATE= effects, then list them in the NONNEGLIGIBLE= effects. Thenonnegligible-effects
listed must be enclosed within parentheses.

You can use NONNEG or N for the keyword NONNEGLIGIBLE.

For example, suppose that you want to construct a fraction of a24 design in order to estimate
the main effects of the four factors. To specify the model, simply list the main effects with the
ESTIMATE= option, since these are the effects of interest. Furthermore, if you consider the two-factor
interactions to be signi�cant but are not interested in estimating them, then list these interactions with
the NONNEGLIGIBLE= option.

See Example 7.8 for an example that uses the ESTIMATE= option. See “Theory of Orthogonal
Designs” on page 687 for details about how the FACTEX procedure interprets the model and derives
an appropriate confounding scheme.

RESOLUTION=r

RESOLUTION=MAXIMUM
speci�es the resolution of the design. The resolution numberr must be a positive integer greater than
or equal to 3. The interpretation ofr is as follows:

• If r is odd, then the effects of interest are taken to be those of order.r � 1/=2 or less.

• If r is even, then the effects of interest are taken to be those of order.r � 2/=2 or less, and the
nonnegligible effects are taken to be those of orderr=2 or less.

If you specify RESOLUTION=MAXIMUM, the FACTEX procedure searches for a design with the
highest resolution that satis�es the SIZE statement requirements.

You can use RES or R for the keyword RESOLUTION and MAX for MAXIMUM.

For more information about design resolution, see the section “Resolution” on page 678. For an
example ofmodel speci�cationthat uses the RESOLUTION=r option, see the section “Example of a
Half-Fraction Factorial Design” on page 614. For an example of the RESOLUTION=MAX option, see
the section “Example of a Full Factorial Design in Two Blocks” on page 612.

The following options for the MODEL statement are available:

MAXCLEAR < ( d) >
requests a search for a design that maximizes the number of clear interactions. Clear interactions are
those which are not aliased with any other effects that are either required to be estimable or assumed to
be nonnegligible. Specifying (d) immediately after the MAXCLEAR option requests a search for a
maximum-clarity design that involves interactions up to orderd . The default value ford is determined
automatically from the model, as it is for the ALIASING option in the EXAMINE statement, as
follows:

• If you specify the model with a resolution numberr in the MODEL statement, thend is the
integer part of.r C 1/=2.

• If you specify the model with a list of effects in the MODEL statement, thend is the larger of the
following, where main effects have order 1, two-factor interactions have order 2, and so on:

– one plus the largest order of an effect to be estimated

– the largest order of an effect considered to be nonnegligible

For details about MaxClear designs, see the section “MaxClear Designs” on page 685.

MINABS < (d) >
requests a search for a design that has minimum aberration. Specifying (d) immediately after the
MINABS option requests a search for a minimum aberration design that involves interactions up to
orderd . The default value ford is determined automatically from the model as follows:

• If you specify the model with a resolution numberr in the MODEL statement, thend D r C 2.

• If you specify the model with a list of effects in the MODEL statement, thend is the larger of the
following, where main effects have order 1, two-factor interactions have order 2, and so on:

– three plus twice the largest order of an effect to be estimated

– one plus twice the largest order of an effect considered to be nonnegligible

See the section “Minimum Aberration” on page 684 for more information. For an example of the
MINABS option, see Example 7.11.

Examples of the MODEL Statement
Suppose you specify a design with the following FACTORS statement, where the number of factorsf can be
replaced with a number:

factors x1-x f ;

Then Table 7.6 lists equivalent ways to specify common models.

Table 7.6 Equivalent of Model Speci�cations

RES= option EST= and NONNEG= options

model res=3 model est=(x1-x+f);
model res=4 model est=(x1-x+ f) nonneg=(x1|x2|x3|+...+|x+f+@2);
model res=5 model est=(x1|x2|x3|+...+|x+ f+@2);

The resolution speci�cation is more concise than the effects speci�cation and is also more ef�cient in an
algorithmic sense. To decrease the time required to �nd a design, particularly for designs with a large number
of factors, you should specify your model that uses the RESOLUTION= option rather than listing the effects.
For more on interpreting the resolution number, see the section “Resolution” on page 678.

OUTPUT Statement

OUTPUT OUT= SAS-data-set < options > ;

You use the OUTPUT statement to save a design in an output data set. Optionally, you can use the OUTPUT
statement to modify the design by specifying values to be output for factors, creating new factors, randomizing
the design, and replicating the design. You specify the output data set as follows:

OUT=SAS-data-set
gives the name of the output data set in which the design is saved. Note that OUT= is required.

options
You can use theoptionsto

• recode the values for design factors

• recode the values for the block variable

• replicate the entire design

• replicate each point of the design

• randomize the design

• create derived factors based on the original factors

The following list describes the precedingoptions:

Recode Design Factors
By default, the output data set contains a variable for each factor in the design coded with standard values, as
follows:

• For factors with 2 levels (q = 2), the values are –1 and +1.

• For factors with 3 levels (q = 3), the values are –1, 0, and +1.

• For factors withq levels (q > 3), the values are0; 1; 2; : : : q� 1.

You can recode the levels of the factor from the standard levels to levels appropriate for your situation.

For example, suppose that you want to recode a three-level factorial design from the standard levels –1, 0,
and +1 to the actual levels. Suppose the factors are pressure (Pressure) with character levels, agitation rate
(Rate) with numeric levels, and temperature (Temperature) with numeric levels. You can use the following
statement to recode the factor levels and save the design in a SAS data set named RECODE:

output out=recode Pressure cvals=(�low� �medium� �high�)
Rate nvals=(20 40 60)
Temperature cvals=(100 150 200);

The general form ofoptionsto recode factors is as follows:

factor-name NVALS= ( level1 level2 . . . levelq)

or

factor-name CVALS=(`level1' `level2' . . . `levelq')

where

factor-name gives the name of the design factor.

NVALS= lists new numeric levels for design factors.

CVALS= lists new character levels for design factors. Each string can be up to 40 characters long.

When recoding a factor, the NVALS= and CVALS= options map the �rst value listed to the lowest value for
the factor, the second value listed to the next lowest value, and so on. If you rename and recode a factor, the
type and length of the new variable are determined by whether you use the CVALS= option (character variable
with length equal to the longest string) or the NVALS= option (numeric variable). For more information
about recoding a factor, see “Factor Variable Characteristics in the Output Data Set” on page 678.

Recode Block Factor
If the design uses blocking, the output data set automatically contains a block variable named BLOCK, and
for a design withb blocks, the default values of the block variable are1; 2; : : : b. You can rename the block
variable and optionally recode the block levels from the default levels to levels appropriate for your situation.

For example, for a design arranged in four blocks, suppose that the block variable is the day of the week
(Day) and that the four block levels of character type areMon, Tue, Wed, andThu. You can use the following
statement to rename the block variable, recode the block levels, and save the design in a SAS data set named
RECODE:

output out=recode blockname=Day cvals=(�Mon��Tue��Wed��Thu�);

The general form ofoptionsto change the block variable name or change the block levels is as follows:

BLOCKNAME= block-name < NVALS= ( level1 level2 . . . levelb) >

or

BLOCKNAME= block-name < CVALS= (`level1' `level2' . . . `levelb') >

where

block-name gives a new name for the block factor.

NVALS= lists new numeric levels for the block factor. For details, see the section “Recode Design
Factors” on page 629.

CVALS= lists new character levels for the block factor. For details, see the section “Recode Design
Factors” on page 629.

Note that you can simply rename the block variable by using only the BLOCKNAME= option, without using
the NVALS= and CVALS= options.

Replicate Entire Design
DESIGNREP=c

DESIGNREP=SAS-data-set
replicates the entire design. Specify DESIGNREP=c to replicate the designc times, wherec is an
integer. Alternatively, you can specify a SAS data set with the DESIGNREP option. In this case, the
design is replicated once for each point in the DESIGNREP= data set, and the OUT= data set contains
the variables in the DESIGNREP= data set in addition to the design variables.

In mathematical notation, the OUT= data set is the direct product of the DESIGNREP= data set and
the design. If the design isa and the DESIGNREP= data set isb, then the OUT= data set isb 
 a,
where
 denotes the direct product.

For details, see the section “Replication” on page 681. For illustrations of the difference between the
DESIGNREP= and POINTREP= options, see Example 7.6 and Example 7.7.

Replicate Design Point
POINTREP=p

POINTREP=SAS-data-set
replicates each point of the design. Specify POINTREP=p to replicate each design pointp times,
wherep is an integer. Alternatively, you can specify a SAS data set with the POINTREP= option. In
this case, the POINTREP= data set is replicated once for each point in the design and the OUT= data
set contains the variables in the POINTREP= data set in addition to the design variables.

In mathematical notation, the OUT= data set is the direct product of the design and the POINT= data
set. If the design isa and the POINTREP= data set isb, then the OUT= data set isa 
 b, where

denotes the direct product.

For details, see the section “Replication” on page 681. For illustrations of the difference between the
DESIGNREP= and POINTREP= options, see Example 7.6 and Example 7.7.

Randomize Design
RANDOMIZE < (u) > < NOVALRAN >

randomizes the design. See the section “Randomization” on page 679 for details. The following
optionsare available:

(u)
speci�es an integer used to start the pseudo-random number generator for randomizing the design.
The value ofu must be enclosed in parentheses immediately after the keyword RANDOMIZE. If
you do not specify a seed, or if you specify a value less than or equal to zero, by default the seed
is generated from reading the time of day from the computer's clock.

NOVALRAN
prevents the randomization of theoretical factor levels to actual levels. The randomization of run
order is still performed.

Create Derived Factors
You can createderived factorsbased on the joint values of a set of the design factors. Each distinct
combination of levels of the design factors corresponds to a single level for the derived factor. Thus, when
you create a derived factor fromk design factors, each withq levels, the derived factor hasqk levels. Derived
factors are useful when you create mixed-level designs; see Example 7.8 for an example. See the section
“Structure of General Factorial Designs” on page 687 for information about how the levels of design factors
are mapped into levels of the derived factor. The general form of theoptionfor creating derived factors is

[ design-factors]= derived-factor < NVALS= ( list-of-numbers) >

or

[ design-factors]= derived-factor < CVALS= (`string1' `string2' . . . `stringn') >

where

design-factors gives names of factors currently in the design. These factors are combined to create the
new derived factor.

derived-factor gives a name to the new derived factor. This name must not be used in the design.

NVALS= lists new numeric levels for the derived factor.

CVALS= lists new character levels for the derived factor. See the section “Recode Design Factors”
on page 629 for details.

If you create a derived factor and do not use the NVALS= or CVALS= option to assign levels to the derived
factor, the FACTEX procedure assigns the values0; 1; : : : ; qk � 1, where the derived factor is created fromk
design factors, each withq levels. In general, the CVALS= or NVALS= list for a derived factor must contain
qk values.

The following statement gives an example of creating a derived factor and then renaming the levels of the
factor:

output out=new [A1 A2]=A cvals=(�A� �B� �C� �D�);

This statement converts two 2-level factors (A1 andA2) into one 4-level factor (A), which has the levelsA, B,
C, andD.

SIZE Statement

SIZE size-speci�cation ;

You use the SIZE statement to specify the size of the design, which is the number of runs in the design. The
SIZE statement is required for designs of less than a full replicate (for example, fractional factorial designs).
By default, the design consists of one full replication of all possible combinations of the factors.

The two simplest explicitsize-speci�cationsthat you can use are

• DESIGN=n, which speci�es the number of runs (n) in the design

• FRACTION=h, which speci�es1=hof one full replicate

Use only one of these two options. In all, there are six mutually exclusivesize-speci�cationsthat you can use,
as described by the following list:

DESIGN=n
speci�es the actual number of runs in the design. The number of runs must be a power of the number
of levelsq for the factors in the design. (See the NLEV= option.) If the last FACTORS statement does
not contain the NLEV= option, thenq = 2 by default, and as a result,n must be a power of 2. For an
example, see Example 7.1.

FRACTION=h
speci�es the fraction of one full replication of all possible combinations of the factors. For instance,
FRACTION=2 speci�es a half-fraction, FRACTION=4 speci�es a quarter-fraction, and so on. In
general, FRACTION=h speci�es a design with1=hof the runs in a full replicate. If the design hasf
factors, each withq levels, then the size of the design isqf =h. If you use FRACTION=h, h must be a
power ofq. See Example 7.4.

NRUNFACS=m
speci�es the number of run-indexing factors in the design. The design contains one run for each
possible combination of the levels of the run-indexing factors. Run-indexing factors are the �rstm
factors for a design inqm runs. All possible combinations of the levels of the run-indexing factors
occur in the design. As a result, if each factor hasq levels, the number of runs in the design isqm .
For details about run-indexing factors, see “Types of Factors” on page 676 and “Structure of General
Factorial Designs” on page 687.

DESIGN=MINIMUM

FRACTION=MAXIMUM

NRUNFACS=MINIMUM
constructs a design with the minimum number of runs (no larger than one full replicate) given all of
the other characteristics of the design. In other words, the design size is optimized. You cannot specify
this option if you specify any of the block size optimization features in the BLOCKS statement (see
NBLKFACS=MAXIMUM).

Equivalence of Speci�cations
The three explicitsize-speci�cationsare related to each other, as demonstrated by the following example.
Suppose you want to construct a design for 11 two-level factors in 128 runs. Since128 D 211=16 D 27,
three equivalent size speci�cations for this design are

size design=128;
size fraction=16;
size nrunfacs=7;

Advanced Examples

Example 7.1: Completely Randomized Design

NOTE : SeeA Completely Randomized Designin the SAS/QC Sample Library.

An experimenter wants to study the effect of cutting speed (Speed) on the surface �nish of a component. He
considers testing the components at �ve levels of cutting speed (100, 125, 150, 175, and 200) and decides to
test �ve components at each level.

The design used is a single-factor completely randomized design with �ve levels and 25 runs. The following
statements generate the required design:

proc factex;
factors Speed / nlev=5;
size design=25;
output out=SurfaceExperiment randomize(713)

Speed nvals=(100 125 150 175 200);
run;
proc print data=SurfaceExperiment;
run;

The RANDOMIZE option in the OUTPUT statement randomizes the run order; the random seed, 713 here,
is optional. The design saved in the data setSurfaceExperiment is displayed in Output 7.1.1.

Output 7.1.1 A Completely Randomized Design

If you are working through this example on your computer, you might �nd a different run order in your
output. This is due to the difference in the seed value of the random number generator. You can specify a
seed value with the RANDOMIZE option. For syntax, see “Randomize Design” on page 632.

Example 7.2: Resolution 4 Augmented Design

NOTE : SeeResolution IV Augmented Designin the SAS/QC Sample Library.

Box, Hunter, and Hunter (1978) describe an injection molding experiment that involves eight 2-level factors:
mold temperature (Temp), moisture content (Moisture), holding pressure (HoldPress), cavity thickness
(Thick), booster pressure (BoostPress), cycle time (Time), screw speed (Speed), and gate size (Gate).

The design used has 16 runs and is of resolution 4; it is often denoted as28� 4
IV . You can generate this design,

shown in Output 7.2.1, with the following statements:

proc factex;
factors Temp Moisture HoldPress Thick

BoostPress Time Speed Gate;
size design=16;
model resolution=4;
examine design aliasing;

run;

The FACTORS statement lists factor names. The option DESIGN=16 of the SIZE statement speci�es the
design size. The RESOLUTION=4 speci�es the resolution of the design. The EXAMINE statement lists
points and aliasing.

Output 7.2.1 A 28� 4
IV Design

The alias structure is shown in Output 7.2.2.

Output 7.2.2 Alias Structure for a 28� 4
IV Design

Subsequent analysis of the data collected for the design suggests thatHoldPress andBoostPress have
statistically signi�cant effects. There also seems to be signi�cant effect associated with the sum of the aliased
two-factor interactionsTemp*BoostPress, Moisture*Time, HoldPress*Speed, andThick*Gate. This chain of
confounded interactions is identi�ed in Output 7.2.2.

A few runs can be added to the design to distinguish between the effects due to these four interactions. You
simply need a design in which these four effects are estimable, regardless of all other main effects and
interactions. For example, the following statements generate a suitable set of runs:

proc factex nocheck;
factors Temp Moisture HoldPress Thick

BoostPress Time Speed Gate;
model estimate=(Temp * BoostPress

Moisture * Time
HoldPress * Speed
Thick * Gate);

size design=min;
examine design aliasing(2);

run;

The DESIGN=MIN option directs FACTEX to search for the smallest design that allows all four interactions
to be estimated. Eight runs are required: see Output 7.2.3.

Output 7.2.3 Additional Runs to Resolve Ambiguities

Output 7.2.4 shows the alias structure of the additional eight runs. Note that the alias chain of interest

Temp* BoostPress=Moisture * Time=HoldPress * Speed=Thick * Gate

from the original design is broken: in this new set of runs, these four interactions are aliased with main
effects and with other two-factor interactions, but they are unaliased with each other. Therefore, when these
four runs are added to the original 16 runs, the main effects of the eight factors plus the four two-factor
interactions that were originally aliased with each other can all be estimated with the 20 runs.

Output 7.2.4 Alias Structure of the Additional Experiment

Example 7.3: Factorial Design with Center Points

NOTE : SeeA Factorial Design with Center Pointsin the SAS/QC Sample Library.

Factorial designs that involve two levels are the most popular experimental designs. For two-level designs,
it is assumed that the response is close to linear over the range of the factor levels. To check for curvature
and to obtain an independent estimate of error, you can replicate points at the center of a two-level design.
Adding center points to the design does not affect the estimates of factorial effects.

To construct a design with center points, you �rst create a data set with factorial points by using the FACTEX
procedure and then augment it with center points by using a simple DATA step. The following example
illustrates this technique.

A researcher is studying the effect of three 2-level factors—current (Current), voltage (Voltage), and time
(Time)—by conducting an experiment using a complete factorial design. The researcher is interested in
studying the overall curvature over the range of factor levels by adding four center points.

You can construct this design in two stages. First, create the basic23 design with the following statements:

proc factex;
factors Current Voltage Time;
output out=Factorial

Current nvals=(12 28)
Voltage nvals=(100 200)
Time nvals=(50 60);

run;

Next, create the center points and append to the basic design as follows:

data Center(drop=i);
do i = 1 to 4;

Current = 20;
Voltage = 150;
Time = 55;
output;

end;
data CPDesign;

set Factorial Center;
run;

proc print data=CPDesign;
run;

The design saved in the data setCPDesign is displayed in Output 7.3.1. Observations 1 to 8 are the factorial
points, and observations 9 to 12 are the center points.

Output 7.3.1 A 23 Design with Four Center Points

Example 7.4: Fold-Over Design

NOTE : SeeA Fold-Over Designin the SAS/QC Sample Library.

Folding overa fractional factorial design is a method for breaking the links between aliased effects in a
design. Folding over a design means adding a new fraction identical to the original fraction except that the
signs of all the factors are reversed. The new fraction is called afold-overdesign. Combining a fold-over
design with the original fraction converts a design of odd resolutionr into a design of resolutionr + 1. 1 For
example, folding over a resolution 3 design yields a resolution 4 design. You can use the FACTEX procedure
to construct the original design fraction and a DATA step to generate the fold-over design.

Consider a1
8 fraction of a26 factorial design with factorsA, B, C, D, E, andF. The following statements

construct a26� 3
III design:

proc factex;
factors A B C D E F;
size fraction=8;
model resolution=3;
examine aliasing;
output out=Original;

run;

title �Original Design�;
proc print data=Original;
run;

The option FRACTION=8 of the SIZE statement speci�es a1
8 fraction of a complete factorial—that is, 8

(=1
8 26). The design, which is saved in the data setOriginal, is displayed in Output 7.4.1.

1This is not true if the original design has even resolution.

Output 7.4.1 A 26� 3
III Design

Since the design is of resolution 3, the alias structure in Output 7.4.2 indicates that all the main effects are
confounded with the two-factor interactions.

Output 7.4.2 Alias Structure for a 26� 3
III Design

To separate the main effects and the two-factor interactions, augment the original design with a 1/8 fraction
in which the signs of all the factors are reversed. The combined design (original design and fold-over design)
of resolution 4 breaks the alias links between the main effects and the two-factor interactions. The fold-over
design can be created by using the following DATA step:

data FoldOver;
set Original;
A=-A; B=-B; C=-C;
D=-D; E=-E; F=-F;

run;
title �Fold-Over Design�;
proc print data=FoldOver;
run;

Here, the DATA step creates the fold-over fraction by reversing the signs of the values of the factors in the
original fraction. The fold-over design is displayed in Output 7.4.3.

Output 7.4.3 A 26� 3
III Design with Signs Reversed

Example 7.5: Randomized Complete Block Design

NOTE : SeeA Randomized Complete Block Designin the SAS/QC Sample Library.

In a randomized complete block design (RCBD), each level of a “treatment” appears once in each block, and
each block contains all the treatments. The order of treatments is randomized separately for each block. You
can create RCBDs with the FACTEX procedure.

Suppose you want to construct an RCBD with six treatments in four blocks. To test each treatment once in
each block, you need 24 experimental units. The following statements construct the randomized complete
block design shown in Output 7.5.1:

proc factex;
factors Block / nlev=4;
output out=Blocks Block nvals=(1 2 3 4) randomize(12345);

run;
factors Treatment / nlev=6;
output out=RCBD

designrep=Blocks
randomize(54321)
Treatment cvals=(�A� �B� �C� �D� �E� �F�);

run;
quit;
proc print data=RCBD;
run;

Note that the order of the runs within each block is randomized and that the blocks (1, 2, 3, and 4) are in a
random order.

Output 7.5.1 A Randomized Complete Block Design

Example 7.6: Two-Level Design with Design Replication and Point
Replication

NOTE : SeeA Two-Level Design with Replicationin the SAS/QC Sample Library.

You can replicate a design to obtain an independent estimate of experimental error or to estimate effects more
precisely. There are two ways you can replicate a design with the FACTEX procedure: you can replicate
the entire design with the DESIGNREP= option, or you can replicate each point in the design with the
POINTREP= option. The following example illustrates the difference.

A process engineer is conducting an experiment to study the shrinkage of an injection-molded plastic
component. The engineer chooses to determine the effect of the following four factors, each at two levels:
holding pressure (Pressure), molding temperature (Temperature), cooling time (Time), and injection velocity
(Velocity).

The design used is a half-fraction of a24 factorial design, denoted as24� 1
IV . The following statements

construct the design:

proc factex;
factors Pressure Temperature Time Velocity;
size fraction=2;
model res=max;
output out=Unreplicated;

run;
proc print data=Unreplicated;
run;

The design, saved in the data setUnreplicated), is shown in Output 7.6.1.

Output 7.6.1 Unreplicated Design

To obtain a more precise estimate of the experimental error, the engineer decides to replicate the entire design
three times. The following statements generate a24� 1

IV design with three replicates in 24 runs:

proc factex;
factors Pressure Temperature Time Velocity;
size fraction=2;
model res=max;
output out=Replicated designrep=3;

run;
proc print data=Replicated;
run;

The design, saved in the data setReplicated, is displayed in Output 7.6.2.

Output 7.6.2 Design Replication

The �rst replicate comprises observations 1 to 8, the second replicate comprises observations 9 to 16, and the
third replicate comprises observations 17 to 24.

Now, instead of replicating the entire design, suppose the engineer decides to replicate each run in the design
three times. The following statements construct a24� 1

IV design in 24 runs with point replication:

proc factex;
factors Pressure Temperature Time Velocity;
size fraction=2;
model res=max;
output out=PointReplicated pointrep=3;

run;
proc print data=PointReplicated;
run;

The design, saved in the data setPointReplicated, is displayed in Output 7.6.3. The �rst design point is
replicated three times (observations 1–3), the second design point is replicated three times (observations 4–6),
and so on.

Output 7.6.3 Point Replication

Note the difference in the arrangement of the designs created by using design replication (Output 7.6.2) and
point replication (Output 7.6.3). In design replication, the original design is replicated a speci�ed number of
times; but in point replication, each run in the original design is replicated a speci�ed number of times. See
the section “Replication” on page 681 for more information on design replication.

Example 7.7: Mixed-Level Design Using Design Replication and Point
Replication

NOTE : SeeA Mixed-Level Design Using Replicationin the SAS/QC Sample Library.

Orthogonal factorial designs are most commonly used at the initial stages of experimentation. At these stages,
it is best to experiment with as few levels of each factor as possible in order to minimize the number of
runs required. Thus, these designs usually involve only two levels of each factor. Occasionally some factors
naturally have more than two levels of interest—different types of seed, for instance.

You can create designs for factors with different numbers of levels simply by taking the crossproduct of
component designs in which the factors all have the same numbers of levels—that is, replicating every run of
one design for each run of the other. (See Example 7.14.) All estimable effects in each of the component

designs, in addition to all generalized interactions between estimable effects in different component designs,
are estimable in the crossproduct; refer to Section 3 of Chakravarti (1956).

This example illustrates how you can construct a mixed-level design by using the OUTPUT statement with
the POINTREP= option or the DESIGNREP= option to take the crossproduct between two designs.

Suppose you want to construct a mixed-level factorial design for two 2-level factors (A andB) and one 3-level
factor (C) with 12 runs. The following SAS statements produce a complete3 � 22 factorial design by using
design replication:

proc factex;
factors A B;
output out=ab;

run;
factors C / nlev=3;
output out=DesignReplicated designrep=ab;

run;
proc print data=DesignReplicated;
run;

Output 7.7.1 lists the mixed-level design saved in the data setDesignReplicated.

Output 7.7.1 3 � 22 Mixed-Level Design Using Design Replication

You can also create a mixed-level design for the preceding factors by using the point replication feature of
the FACTEX procedure. The following SAS statements produce a complete22 � 3 factorial design by using
point replication:

proc factex;
factors A B;
output out=ab;

run;
factors C / nlev=3;
output out=PointReplicated pointrep=ab;

run;
proc print data=PointReplicated;
run;

Output 7.7.2 lists the mixed-level design saved in the data setPointReplicated.

Output 7.7.2 22 � 3 Mixed-Level Design Using Point Replication

Note the difference between the designs in Output 7.7.1 and Output 7.7.2. In design replication, the mixed-
level design is given byAB 
 C, while for point replication the mixed-level design is given byC 
 AB,
where
 denotes the direct product. In design replication, you can view the DESIGNREP= data set as nested
outsidethe design, while in point replication, you can view the POINTREP= data set as nestedinsidethe
design.

Example 7.8: Mixed-Level Design Using Pseudo-Factors

NOTE : SeeMixed-Level Designs Using Pseudo-Factorsin the SAS/QC Sample Library.

If the numbers of levels for the factors of the mixed-level design are all powers of the same prime power
q, you can construct the design by usingpseudo-factors, where the levels ofk q-level pseudo-factors are
associated with the levels of a singlederived factorwith qk levels. Refer to Section 5 of Chakravarti (1956)
and see the section “Types of Factors” on page 676 for details.

For example, the following statements create a design for one 4-level factor (A) and three 2-level factors (B,
C, andD) in 16 runs (a half replicate):

proc factex;
factors A1 A2 B C D;
model estimate =(B C D A1|A2 )

nonnegligible=(B|C|D@2 A1|A2|B A1|A2|C A1|A2|D);
size design=16;
output out=DesignA [A1 A2]=A cvals = (�A� �B� �C� �D�);

run;
proc print;

var A B C D;
run;

The levels of two 2-level pseudo-factors (A1 andA2) are used to represent the four levels ofA. Hence the
three degrees of freedom associated withA are given by the main effects ofA1 andA2 and their interaction
A1*A2, and you can thus refer to (A1|A2) as the main effect ofA.

The MODEL statement speci�es that the main effects of all factors are to be estimable, and that all of the
two-factor interactions betweenB, C, andD, in addition to the interactions between each of these and (A1|A2),

are to be nonnegligible. As a result, the mixed-level design has resolution 4. The design is saved in the data
setDesignA, combining the levels of the two pseudo-factors,A1 andA2, to obtain the levels of the 4-level
factorA. The data setDesignA is listed in Output 7.8.1.

Output 7.8.1 4 � 23 Design of Resolution 4 in 16 Runs

Example 7.9: Mixed-Level Design by Collapsing Factors

NOTE : SeeMixed-Level Design with Collapsing Factorsin the SAS/QC Sample Library.

You can construct a mixed-level design bycollapsingfactors—that is, by replacing a factor withn levels by a
factor withm levels, wherem < n . Orthogonality is retained in the sense that estimates of different effects
are uncorrelated, although not all estimates have equal variance; refer to Section 6 of Chakravarti (1956).
This method has been used by Addelman (1962) to derive main effects plans for factors with mixed numbers
of levels and by Margolin (1967) to construct plans that consider two-factor interactions.

You can use the value speci�cation in the OUTPUT statement as a convenient tool for collapsing factors. For
example, the following statements create a 27-run design for two 2-level factors (x1 andx2) and two 3-level
factors (x3 andx4) such that all main effects and two-factor interactions are uncorrelated:

proc factex;
factors x1-x4 / nlev = 3;
size design=27;
model r=4;
output out=MixedLevel x1 nvals=(-1 1 -1)

x2 nvals=(-1 1 -1);
run;
proc print data=MixedLevel;
run;

The mixed-level design is a three-quarter fraction with resolution 5; refer to Margolin (1967). The design is
displayed in Output 7.9.1.

Output 7.9.1 22 � 32 Design of Resolution V in 27 Runs

Example 7.10: Hyper-Graeco-Latin Square Design

NOTE : SeeHyper-Graeco-Latin Squarein the SAS/QC Sample Library.

A q � q Latin square is an arrangement ofq symbols, each repeatedq times, in a square of sideq such
that each symbol appears exactly once in each row and in each column. Such arrangements are useful as
designs forrow-and-columnexperiments, where it is necessary to balance the effects of twoq-level factors
simultaneously.

A Graeco-Latin square is actually a pair of Latin squares; when superimposed, each symbol in one square
occurs exactly once with each symbol in the other square. The following is an example of a5� 5 Graeco-Latin
square, where Latin letters are used for the symbols of one square and Greek letters are used for the symbols
of the other:

A� B� C 
 D� E�
B
 C � D� E� A�
C� D� E� A
 B�
D� E
 A� B� C�
E� A� B� C� D

Wheneverq is a power of a prime number, you can construct up toq – 1 squares, each withq symbols that
are balanced over all the other factors. The result is called ahyper-Graeco-Latin squareor a complete set of
mutually orthogonalLatin squares. Such arrangements can be useful as designs (refer to Williams 1949), or
they can be used to construct other designs.

Whenq is a prime power, hyper-Graeco-Latin squares are straightforward to construct with the FACTEX
procedure. This is because a complete set ofq – 1 mutually orthogonalq � q Latin squares is equivalent to a
resolution 3 design forq + 1 q-level factors inq2 runs, where two of the factors index rows and columns and
each of the remaining factors indexes the treatments of one of the squares.

For example, the following statements generate a complete set of three mutually orthogonal4 � 4 Latin
squares, with rows indexed by the factorRow, columns indexed by the factorColumn, and the treatment
factors in the respective squares indexed byt1, t2, andt3. The �rst step is to construct a resolution 3 design
for �ve 4-level factors in 16 runs.

proc factex;
factors Row Column t1-t3 / nlev=4;
size design=16;
model resolution=3;
output out=OrthArray t1 cvals=(�A� �B� �C� �D�)

t2 cvals=(�A� �B� �C� �D�)
t3 cvals=(�A� �B� �C� �D�);

run;

data _null_;
array t{3} $ t1-t3;
array s{4} $ s1-s4; / * Buffer for holding each row * /
file print; / * Direct printing to output screen * /
do square=1 to 3;

put "Square " square ":";
n = 1;
do r=1 to 4;

do c=1 to 4;
set OrthArray point=n; n=n+1;
s{c}=t{square};

end;
put " " s1-s4;

end;
put;

end;
stop;

run;

In most cases, the form that appears in the output data setOrthArray is most useful. The form that usually
appears in textbooks is displayed in Output 7.10.1, which can be produced by using a simple DATA step (not
shown here).

Output 7.10.1 Hyper-Graeco-Latin Square

Example 7.11: Resolution 4 Design with Minimum Aberration

NOTE : SeeA Res IV Design with Minimum Aberrationin the SAS/QC Sample Library.

If a design has resolution 4, then you can simultaneously estimate all main effects andsometwo-factor
interactions. However, not all resolution 4 designs are equivalent; you might be able to estimate more
two-factor interactions with some than with others. Among all resolution 4 designs, a design with the
maximum number of estimable two-factor interactions is said to haveminimum aberration.

For example, if you use the FACTEX procedure to generate a resolution 4 2-level design in 32 runs for seven
factors, you can estimate all main effects and 15 of the 21 two-factor interactions with the design that is
created by default. The following statements create this design and display its alias structure in Output 7.11.1:

proc factex;
factors A B C D E F G;
model resolution=4;
size design=32;
examine aliasing;

run;

Output 7.11.1 Alias Structure for Default 27� 2
IV Design

In contrast, the resolution 4 design given in Table 12.15 of Box, Hunter, and Hunter (1978) is a minimum
aberration design that permits estimation of 18 two-factor interactions, three more than can be estimated with
the default design. The FACTEX procedure constructs the minimum aberration design if you specify the
MINABS option in the MODEL statement, as in the following statements:

proc factex;
factors A B C D E F G;
model resolution=4 / minabs;
size design=32;
examine aliasing;

run;

The alias structure for the resulting design is shown in Output 7.11.2.

Output 7.11.2 Alias Structure for Minimum Aberration 27� 2
IV Design

All of the designs listed in Table 12.15 of Box, Hunter, and Hunter (1978) have minimum aberration. For
most of these cases, the default design constructed by the FACTEX procedure has minimum aberration—that
is, the MINABS option is not required. This is important because the MINABS option forces the FACTEX
procedure to check many more designs, and the search can therefore take longer to run. You can limit the
search time with the TIME= option in the PROC FACTEX statement. In �ve of the cases (210� 6

III , 27� 2
IV ,

28� 3
IV , 29� 4

IV , and210� 3
V ), the MINABS option is required to construct a design with minimum aberration,

and in two cases (29� 5
III , 29� 3

IV ), the NOCHECK option is also required. If the FACTEX procedure is given
suf�cient time to run, specifying both the MINABS and the NOCHECK options always results in a minimum
aberration design. However, with the default search time of 60 seconds, there are three cases (210� 5

IV , 210� 4
IV ,

and211� 5
IV ) for which the FACTEX procedure is unable to �nd the minimum aberration design, even with

both the MINABS and NOCHECK options speci�ed.

Example 7.12: Replicated Blocked Design with Partial Confounding

NOTE : SeeReplicated Blocked Design with Confoundingin the SAS/QC Sample Library.

In an unreplicated blocked design, the interaction effect that is confounded with the block effect cannot
be estimated. You can replicate the experiment so that a different interaction effect is confounded in each
replicate. This enables you to obtain information about an interaction effect from the replicates in which it is
not confounded.

For example, consider a23 design with factorsA, B, andC arranged in two blocks. Suppose you decide
to run four replicates of the design. By constructing the design sequentially, you can choose the effects to
be estimated in each replicate depending on the interaction confounded with the block effect in the other
replicates.

In the �rst replicate, you specify only that the main effects are to be estimable. The following statements
generate an eight-run two-level design arranged in two blocks:

proc factex;
factors A B C;
blocks nblocks=2;
model est=(A B C);
examine confounding aliasing;
output out=Rep1 blockname=block nvals=(1 2);

run;

The alias structure and the confounding scheme are listed in Output 7.12.1. The highest-order interaction
A*B*C is confounded with the block effect. The design, with recoded block levels, is saved in a data set
named REP1.

Output 7.12.1 Confounding Rule and Alias Structure for Replicate 1

If you were to analyze this replicate by itself, you could not determine whether an effect is due toA*B*C or
the block effect. You can construct a second replicate that confounds a different interaction effect with the
block effect. Since the FACTEX procedure is interactive, simply submit the following statements to generate
the second replicate:

model est=(A B C A * B* C);
output out=Rep2

blockname=block nvals=(3 4);
run;

The alias structure and the confounding scheme for the second replicate are listed in Output 7.12.2. The
interactionA*B*C is free of any aliases, but now the two-factor interactionB*C is confounded with the block
effect.

Output 7.12.2 Confounding Rule and Alias Structure for Replicate 2

To estimate the interactionB*C with the third replicate, submit the following statements (immediately after
the preceding statements):

model est=(A B C A * B* C B* C);
output out=Rep3 blockname=block nvals=(5 6);

run;

The alias structure and confounding rules are shown in Output 7.12.3. The interactionB*C is free of aliases,
but the interactionA*C is confounded with the block effect.

Output 7.12.3 Confounding Rule and Alias Structure for Replicate 3

Finally, to estimate the interaction effect A*C with the fourth replicate, submit the following statements:

model est=(A B C A * B* C B* C A* C);
output out=Rep4 blockname=block nvals=(7 8);

run;

The alias structure and confounding rules are displayed in Output 7.12.4.



656 F Chapter 7: The FACTEX Procedure

Output 7.12.4 Confounding Rule and Alias Structure for Replicate 4

When combined, these four replicates give full information on the main effects and three-quarter information
on each of the interactions. The following statements combine the four replicates:

data Combine;
set Rep1 Rep2 Rep3 Rep4;

run;
proc print data=Combine;
run;

The �nal design is saved in the data setCombine. A partial listing of this data set is shown in Output 7.12.5.

Output 7.12.5 Combined Design

Example 7.13: Incomplete Block Design

NOTE : SeeIncomplete Block Designin the SAS/QC Sample Library.

Several important series of balanced incomplete block designs can be derived from orthogonal factorial
designs. One is the series of balanced lattices of Yates (1936); refer to page 396 of Cochran and Cox
(1957). In a balanced lattice, the number of treatmentsv must be the square of a power of a prime number:
v D q2; q D pk , wherep is a prime number. These designs are based on a complete set ofq – 1 mutually
orthogonalq � q Latin squares, which is equivalent to a resolution 3 design forq + 1 q-level factors inq2

runs.

The balanced lattice designs includeq + 1 replicates of the treatments. They are constructed by associating
each treatment with a run in the factorial design, each replicate with one of the factors, and each block

with one of theq values of that factor. For example, the treatments in Block 3 within Replicate 2 are those
treatments that are associated with runs for which factor 2 is set at value 3.

The following statements use this method to construct a balanced lattice design for 16 treatments in �ve
replicates of four blocks each. The construction procedure is based on a resolution 3 design for �ve 4-level
factors in 16 runs.

proc factex;
factors x1-x5 / nlev=4;
size design=16;
model r=3;
output out=a;

run;

In the following DATA step, the incomplete block design is built by using the design saved in the data seta
by the FACTEX procedure:

data b;
keep Rep Block Plot t;
array x{5} x1-x5;
do Rep = 1 to 5;

do Block = 1 to 4;
Plot = 0;
do n = 1 to 16;

set a point=n;
if (x{rep}=Block-1) then do;

t = n;
Plot = Plot + 1;
output;

end;
end;

end;
end;
stop;

run;

For each block within each replicate, the program loops through the run numbers in the factorial design and
chooses those that have theRepth factor equal toBlock–1. These run numbers are the treatments that go into
the particular block.

The design is printed by using a DATA step. Each block of each replicate is built into the variables S1, S2,
S3, and S4, and each block is printed with a PUT statement.

data _null_;
array s{4} s1-s4;
file print;
n = 1;
do r = 1 to 5;

put "Replication " r 1.0 ":";
do b = 1 to 4;

do p = 1 to 4;
set b point=n;
s{Plot} = t;
n = n+1;

end;

put " Block " b 1.0 ":" (s1-s4) (3.0);
end;
put;

end;
stop;

run;

The ARRAY statement creates a buffer for holding each block, and the FILE statement directs the printing to
output screen. The design is displayed in Output 7.13.1.

You can use the PLAN procedure to randomize the block design, as shown by the following statements:

proc plan seed=54321;
factors Rep=5 Block=4 Plot=4 / noprint;
output data=b out=c;

run;
proc sort;

by Rep Block Plot;
run;

The variablePlot indexes the plots within each block. Refer to theSAS/STAT User's Guidefor a general
discussion of randomizing block designs.

Finally, substituteset c for set b in the preceding DATA step. Running this DATA step creates the
randomized design displayed in Output 7.13.2.

Output 7.13.1 A Balanced Lattice

Output 7.13.2 Randomized Design

Example 7.14: Design with Inner Array and Outer Array

NOTE : SeeA Problem In Quality Improvementin the SAS/QC Sample Library.

Byrne and Taguchi (1986) report the use of a fractional factorial design to investigate �tting an elastomeric
connector to a nylon tube as tightly as possible. Their experiment applies the design philosophy of Genichi
Taguchi, which distinguishes between control factors and noise factors.Control factorsare typically those
that the engineer is able to set under real conditions, whilenoise factorsvary uncontrollably in practice
(though within a predictable range).

The experimental layout consists of two designs, one for the control factors and one for the noise factors.
The design for the control factors is called theinner array, and the design for noise factors is called the
outer array. The outer array is replicated for each of the runs in the inner array, and a performance measure
(“signal-to-noise ratio”) is computed over the replicate. The performance measure thus re�ects variation due
to changes in the noise factors. You can construct such a crossproduct design with the replication options in
the OUTPUT statement of the FACTEX procedure, as shown in this example.

Researchers identi�ed the following four control factors that were thought to in�uence the amount of force
required to pull the connector off the tube:

• interference (Interference), de�ned as the difference between the outer width of the tubing and the
inner width of the connector

• connector wall thickness (ConnectorWall)

• depth of insertion (InsertDepth) of the tubing into the connector

• amount of adhesive (Glue) in the connector pre-dip

Researchers also identi�ed the following three noise factors related to the assembly:

• amount of time (Time) allowed for assembly

• temperature (Temperature))

• relative humidity (Humidity)

Three levels were selected for each of the control factors, and two levels were selected for each of the noise
factors.

The following statements construct the 72-run design used by Byrne and Taguchi (1986). First, an eight-run
outer array for the three noise factors is created and saved in the data set OUTERARY.

proc factex;
factors Time Temperature Humidity;
output out=OuterArray Time nvals=( 24 120)

Temperature nvals=( 72 150)
Humidity nvals=(0.25 0.75);

run;

Next, a nine-run inner array (design of resolution 3) is chosen for the control factors. The POINTREP option
in the OUTPUT statement replicates the eight-run outer array in the data set OUTERARY for each of the
nine runs in the inner array and saves the �nal design containing 72 runs in the data setDesign.

proc factex;
factors Interference ConnectorWall InsertDepth Glue /

nlev=3;
size design=9;
model resolution=3;
output out=Design pointrep=OuterArray

Interference cvals=(�Low� �Medium� �High� )
ConnectorWall cvals=(�Thin� �Medium� �Thick� )
InsertDepth cvals=(�Shallow� �Deep� �Medium�)
Glue cvals=(�Low� �High� �Medium�);

run;

The �nal design is listed in Output 7.14.1. Main effects of each factor can be estimated free of each other but
are confounded with two-factor interactions.

Output 7.14.1 Design for Control Factor and Noise Factors

Output 7.14.1 continued

Note that the levels ofInsertDepth andGlue are listed in the OUTPUT statement in a nonstandard order so
that the design produced by the FACTEX procedure matches the design of Byrne and Taguchi (1986). The
order of assignment of levels does not affect the properties of the resulting design. Furthermore, the design
can be randomized with the RANDOMIZE option in the OUTPUT statement.

Byrne and Taguchi (1986) indicate that a smaller outer array with only four runs would have been suf�cient.
You can generate this design (not shown here) by modifying the statements in this example; speci�cally, add
the following SIZE and MODEL statements:

size design=4;
model resolution=3;

In their analysis of the data from the experiment based on the smaller design, Byrne and Taguchi (1986)
note several interesting interactions between control and noise factors. However, since the inner array is of
resolution 3, it is impossible to say whether interesting interactions exist between the control factors. In other
words, you cannot determine whether an effect is due to an interaction or to the main effect with which it is
confounded.

One alternative is to begin with a design of resolution 4. Two-factor interactions remain confounded with one
another, but they are free of main effects. Moreover, further experimentation can be carried out to distinguish



664 F Chapter 7: The FACTEX Procedure

between confounded interactions that seem important. To determine the optimal size of this design, submit
the following statements interactively:

proc factex;
factors Interference ConnectorWall InsertDepth Glue /

nlev=3;
model resolution=4;
size design=minimum;

run;

This causes the following message to appear in the SAS log:

NOTE: Design has 27 runs, resolution = 4.

In other words, the smallest resolution 4 design for four 3-level factors has 27 runs, which together with the
eight-run outer array requires 216 runs. Even the smaller four-run outer array requires 108 runs. Both of
these designs are substantially larger than the design originally reported, but the larger designs protect against
the effects of unsuspected interactions.

A second alternative is to begin with only two levels of the control factors. Further experimentation can
then be directed toward exploring the effects of factors determined to be important in this initial stage
of experimentation. Note that NLEV=2 is the default in the FACTORS statement. Submit the following
additional statements:

factors Interference ConnectorWall InsertDepth Glue;
model resolution=4;
size design=minimum;

run;

This causes the following message to appear in the SAS log:

NOTE: Design has 8 runs, resolution = 4.

Thus, as few as eight runs can be used for the inner array. This design is amenable to blocking, whereas the
proposed nine-run design is not. Blocking is an important consideration whenever experimental conditions
can vary over the course of conducting the experiment.

Now, submit the following statements:

size design=8;
blocks size=minimum;

run;

This causes the following message to appear in the SAS log:

NOTE: Design has 8 runs in 4 blocks of size 2,
resolution = 4.

Thus the experiment can be run in blocks as small as two runs.

Example 7.15: Fractional Factorial Split-Plot Designs

NOTE : SeeFractional Factorial Split-Plot Designin the SAS/QC Sample Library.

In split-plot designs, not all factor levels can change from plot to plot. In the simplest split-plot structure,
runs are grouped into whole plots; certain factors are applied to all plots in the whole plot, while others are
applied to individual plots within a whole plot. The two types of factors are termedwhole-plot factorsand
subplot factors, respectively. Split-plot designs are very common in chemical and process industries, where
factors of interest are often applied at different stages of the production process and the �nal measurements
of interest are made on the �nished product. In this case, the different stages of production might give rise to
multiple whole-plot effects.

Suppose you are designing an experiment to measure six factors that affect characteristics of metal wires
sheathed with a certain material. Three of the factors (W1, W2, W3) apply to how the wires themselves are
made, and the other three (S1, S2, S3) apply to the sheathing material. You propose to �rst prepare eight
different batches of wire, making two wires from each batch, and then to prepare the sheathing material
for each wire individually. This describes a standard split-plot experiment with batches of wires forming
whole plots and sheathed wires forming subplots. The following code constructs a resolution 4 design for
this experiment, specifying theWire unit effect in the BLOCKS statement, and then in the UNITEFFECT
statement specifying thatW1, W2, andW3 should be constant withinWire and thatS1, S2, andS3 should
change withinWire. The resulting design is printed, sorted byWire.

proc factex;
factors W1 W2 W3

S1 S2 S3;
size design=16;
blocks units=(Wire=8);
model r=4;
uniteffect Wire / whole=(W1 W2 W3)

sub =(S1 S2 S3);
examine aliasing(units);
output out=WireExperiment1;

run;

proc sort data=WireExperiment1;
by Wire W1-W3 S1-S3;

run;
proc print data=WireExperiment1;
run;

Output 7.15.1 shows the aliasing structure for the design, which indicates that the main effects of the wire
factors are indeed estimated on theWire whole plots and the main effects of the sheath factors are estimated
on the subplots. Interestingly, some of the sheath factor interactions are also confounded with whole plots.



666 F Chapter 7: The FACTEX Procedure

Output 7.15.1 A Split-Plot Design

The �nal design is listed in Output 7.15.2. Notice that the factorsW1, W2, andW3 are constant withinWire,
while S1, S2, andS3 change withinWire.

Output 7.15.2 A Split-Plot Design

To see why theWire factors are constant within wire and the sheath factors change, examine the confounding
rules for the design. The following statements produce the table of confounding rules listed in Output 7.15.3:

proc factex;
factors W1 W2 W3

S1 S2 S3;
size design=16;

blocks units=(Wire=8);
model r=4;
uniteffect Wire / whole=(W1 W2 W3)

sub =(S1 S2 S3);
examine confounding;

run;

Output 7.15.3 Split-Plot Confounding Rules

The terms [i] on the right-hand side of these rules denote plot-indexing pseudo-factors, as discussed in the
section “Split-Plot Designs” on page 686. Note that the wire factorsW1, W2, andW3 are confounded only
with interactions between the �rst three pseudo-factors, the ones identi�ed with the eight levels of theWire
unitfactor. This guarantees that these factors are constant within levels ofWire. By contrast, the confounding
rules for the sheath factorsS1, S2, andS3 each involve the fourth pseudo-factor, so they must change within
levels ofWire.

There are only eight different combinations of the sheath factors, but the previous design requires you to
produce batches of sheath material 16 times, once for each of the two wires to be made from each wire
batch. If instead you propose to make just four batches of sheath material and apply part of each batch to
parts of different batches of wires, the design becomes a row-column design instead of a split-plot design.
Furthermore, suppose that the number of batches rather than the size of each batch is the main cost, so that
you can prepare eight batches of wire and four batches of sheathing material in suf�cient quantity to make 64
different sheathed wires. Since there can be only four different combinations of the three sheathing factors,
each sheathing factor interaction is aliased with a main effect, and thus the design necessarily has resolution
3. All other interactions are estimable free of main effects. The following FACTEX statements create the
design and display the two unit effects with their respective wholeunit factor levels:

proc factex;
factors W1 W2 W3

S1 S2 S3;
size design=64;
blocks units=(Wire=8 Sheath=4);
model r=3;
uniteffect Wire / whole=(W1 W2 W3);
uniteffect Sheath / whole=(S1 S2 S3);
examine aliasing(units);
output out=WireExperiment2;

proc freq data=WireExperiment2;
table Wire * W1* W2* W3 / list nocum nopct;
table Sheath * S1* S2* S3 / list nocum nopct;

run;

Output 7.15.4 A Split-Lot Design: Wire Units

Output 7.15.5 A Split-Lot Design: Sheath Units

The results, listed in Output 7.15.4 and Output 7.15.5, indicate thatW1, W2, andW3 are constant within
Wire andS1, S2, andS3 are constant withinSheath.

Example 7.16: A Design for a Three-Step Process

NOTE : SeeA Design for a Three-Step Processin the SAS/QC Sample Library.

Ramirez and Weisz (2009) discuss an experiment on a multistep milling process with 16 processing factors,
with a single factor applied at the �rst stage, seven more factors at the second stage, and eight more at the
�nal stage. The experiment involves eight �rst-stage runs, eight second-stage runs within each of those, and
again, two to four third-stage runs within each of those, for a total of 128 to 256 total experimental units.
This example explores several different ways to design this experiment, depending on what kinds of effects
are most important.

The following statements request a design of maximum resolution for this split-plot structure. The factors are
listed in macro variables, for ease in specifying them in UNITEFFECT statements. The BLOCK statement
de�nes the unit factors for the �rst two processing stages, with eight runs of each. The two UNITEFFECT
statements then use these unit factors to specify which unit effects correspond to which factors. Finally, the
EXAMINE statement requests that the aliasing structure and the overall modeling summary be displayed to
see how many effects of different orders are estimable and clear. The UNITS suboption of the ALIASING
option includes the unit effect confounding for each alias string in the alias structure.

%let F1 = Z;
%let F2 = A B C D E F G;
%let F3 = P Q R S T U V W;
proc factex;

factors &F1 &F2 &F3;
model r=max;
size design=128;
blocks units=(Step1=8 Step2=8);
uniteffect Step1 / whole=(&F1) sub=(&F2 &F3);
uniteffect Step1 * Step2 / whole=(&F2) sub=( &F3);
examine aliasing(units) summary;

quit;

Output 7.16.1 Aliasing for Default 128-Run Three-Step Design

Output 7.16.1 continued

Output 7.16.2 Modeling Summary for Default 128-Run Three-Step Design

The resulting design has resolution 4, which means that main effects are clear of two-factor interactions but
interactions are aliased with each other. Output 7.16.1 shows which interactions are aliased and also shows
which units are used to estimate them. Note that several interactions betweenStep2 andStep3 factors are
estimated withStep2 units.

As Output 7.16.2 shows, only 30/120=25% of the two-factor interactions (2FI) are estimable and only
15/120=13% of them are clear. If simply protecting the main-effects estimates against potential two-factor
interactions is suf�cient, then this design suf�ces. However, if you want to estimate as many of the two-
factor interactions as possible, then you should look for a MaxClear design. The following statements use
the MAXCLEAR option in the MODEL statement to request a MaxClear design, and they also use the
ORDER=RANDOM(RESTART) option in the PROC FACTEX statement to improve the chances that the
best design is found. For more details about MaxClear designs, see the section “MaxClear Designs” on
page 685.

%let F1 = Z;
%let F2 = A B C D E F G;
%let F3 = P Q R S T U V W;
proc factex order=random(restart seed=1);

factors &F1 &F2 &F3;
model r=max / maxclear;
size design=128;
blocks units=(Step1=8 Step2=8);
uniteffect Step1 / whole=(&F1) sub=(&F2 &F3);
uniteffect Step1 * Step2 / whole=(&F2) sub=( &F3);
examine summary;

quit;

Output 7.16.3 Modeling Summary for MaxClear 128-Run Three-Step Design

The modeling summary results for the MaxClear design are shown in Output 7.16.3. Now 87/120=73% of
the 2FI are estimable and 69/120=58% of them clear. This is a great improvement over the default design,
but if complete estimability of all two-factor interactions is required, more than 128 runs are necessary. The
following statements construct a design in 256 runs, effectively doubling the number of third-stage runs from
two to four:

%let F1 = Z;
%let F2 = A B C D E F G;
%let F3 = P Q R S T U V W;
proc factex;

factors &F1 &F2 &F3;
model r=max;
size design=256;
blocks units=(Step1=8 Step2=8);
uniteffect Step1 / whole=(&F1) sub=(&F2 &F3);
uniteffect Step1 * Step2 / whole=(&F2) sub=( &F3);
examine aliasing(units);

quit;

The aliasing structure (not shown) shows that the resulting design has resolution 5, which means that all main
effects and two factor interactions are estimable free of each other. Even though the required 256 runs means
this is a relatively large experiment, it is still only a tiny fraction of the 65,536 runs required for a complete
factorial design.

Example 7.17: A Strip-Split-Split-Plot Design

NOTE : SeeA Strip-Split-Split-Plot Designin the SAS/QC Sample Library.

Suppose you are designing an experiment for a three-step process running on different machines. One way to
model this is with a row� column strip-split-split-plot structure, with one type of unit,Machine, crossed with a
process that has a split-split-plot structure. The following statements create a resolution 4 design in 11 factors
for this situation, with oneMachine factorMSetting and three, three, and �ve whole plot, split-plot, and
split-split-plot process factors, respectively. The statements also specify that the design's aliasing structure
and modeling summary be displayed, with the unit effect confounding for each alias string included in the
alias structure.

